OPTIMIZACIÓN ESTÁTICA MULTIVARIANTE CON MAPLE

<u>Autor:</u>

Jorge Mauricio Oviedo [1]

Facultad de Ciencias Económicas. Universidad Nacional de Córdoba

Resumen: El presente trabajo provee un conjunto de rutinas capaces de crear nuevos comandos reconocidos por el lenguaje Maple que habilitan al usuario a resolver problemas de optimización en n variables y m restricciones. La generalidad del mismo permite abordar casos de optimización libre y con restricciones de igualdad y desigualdad facultando el cómputo de las condiciones de primer y segundo orden.

[1] joviedo@eco.unc.edu.ar

1- INTRODUCCIÓN

La resolución de problemas de optimización tanto numérica como simbólicamente suele ser una tarea larga, tediosa y engorrosa en el sentido del alto insumo de tiempo que esta demanda. En este sentido, el uso de Softwares para facilitar esta tarea, surge como una idea atractiva pero lastimosamente hasta el momento los paquetes internos provistos por los distintos programas (Maple y Mathematica principalmente) no contienen comandos fáciles y directos que permitan a una persona sin dominio en le manejo de los mismos, resolver estos problemas.
El objetivo de este trabajo es brindar un paquete de optimización que resulte sumamente fácil de implementar a cada problema con el que el usuario se enfrente. Vale destacar que el mismo provee las soluciones tanto numéricas como simbólicas a cada tipo de problemas.
Los problemas que se resuelven mediante este paquete son los siguientes: Optimización Libre en n variables, Optimización con restricciones de igualdad en n variables y m restricciones (m < n) y Optimización con restricciones de desigualdad en n variables y m restricciones.

2- CLASIFICACIÓN DE LOS PROBLEMAS DE OPTIMIZACIÓN

En una primera gran división los problemas de Optimización pueden dividirse:

- **Optimización Estática:** donde el objetivo es hallar el valor numérico de una o varias variables que optimizan el valor de una función en una o varias variables respectivamente. A su vez éstos problemas pueden subclasificarse en:
  - Optimización Libre
  - Optimización sujeto a restricciones de igualdad
  - Optimización sujeto a restricciones de desigualdad

- **Optimización Dinámica:** donde el objetivo es hallar una función o conjunto de funciones que maximicen o minimicen una integral definida. Estos problemas pueden clasificarse en tiempo discreto y continuo por un lado, en problemas determinísticos o aleatorios por otro y dentro de estos en variables aleatorias continuas o discretas. Este tipo de optimización exceden el alcance de este trabajo y se proveerá en un trabajo posterior.

3- OPTIMIZACIÓN LIBRE

Caso 2 variables:
En este problema el objetivo consiste en determinar los valores de x_1 y x_2 que hacen máximo o mínimo el valor de una función objetivo f. Formalmente el problema puede escribirse de la siguiente manera:

$$\boxed{\underset{x_1,x_2}{Max}\ f = f(x_1, x_2)}$$

Donde la condición de primer orden (condición necesaria) viene dada por el siguiente sistema de ecuaciones[2]:

[2] Donde $f_{x_i}(x_1, x_2, \ldots, x_n) = \dfrac{\partial f(x_1, x_2, \ldots, x_n)}{\partial x_i} = \lim_{h \to 0} \dfrac{f(x_1, x_2, \ldots x_i + h, \ldots, x_n) - f(x_1, x_2, \ldots x_i, \ldots, x_n)}{h}$

$$\begin{cases} f_{x_1}(x_1, x_2) = 0 \\ f_{x_2}(x_1, x_2) = 0 \end{cases} \Rightarrow (x_1^*, x_2^*)$$

Geométricamente la condición de primer orden establece que hay que buscar los valores de las variables, es decir el punto del dominio para el cual el plano tangente a la gráfica de *f* es horizontal. Dicho sistema por lo general es no lineal y puede tener varias soluciones es decir pueden existir varios puntos para los cuales el plano tangente a la gráfica de *f* es horizontal. Dichos puntos son denominados puntos críticos o candidatos a óptimos de *f*. ya que los puntos donde *f* presenta plano tangente horizontal pueden ser máximos, mínimos o puntos de inflexión (puntos de silla). Para dicha distinción la condición de segundo orden (condición suficiente) hace referencia a la matriz hessiana (matriz de derivadas segundas) evaluada en cada punto crítico:

$$\mathbf{H}_f(x_1^*, x_2^*) = \begin{pmatrix} f_{x_1 x_1}(x_1^*, x_2^*) & f_{x_1 x_2}(x_1^*, x_2^*) \\ f_{x_2 x_1}(x_1^*, x_2^*) & f_{x_2 x_2}(x_1^*, x_2^*) \end{pmatrix}$$

Dicha matriz provee toda la información necesaria de *f* en las inmediaciones de cada punto por medio de una aproximación de Taylor de segundo orden en el sentido de que avisa si el entorno del punto es cóncavo o convexo. Algebraicamente, para discernir si cada punto crítico es efectivamente un punto que optimiza a *f* se hace referencia la concepto de menores principales de orden n asociado a una matriz cuadrada de orden m (m >= n). Un menor principal de orden n asociado a una matriz de orden m es el determinante que surge de considerar una submatriz de orden n conformada por las primeras n filas y las primeras n columnas. Con esta definición la condición de segundo orden se puede enunciar en términos de los menores principales de la matriz hessiana evaluada en cada punto crítico. En efecto:

$$\textit{Mínimo} \begin{cases} |\mathbf{H}_1| > 0 \\ |\mathbf{H}_2| > 0 \end{cases} \qquad \textit{Máximo} \begin{cases} |\mathbf{H}_1| < 0 \\ |\mathbf{H}_2| > 0 \end{cases}$$

Es decir para que efectivamente un punto crítico (que satisface la condición de primer orden) sea un máximo local de *f* es suficiente que los menores principales asociados a la matriz hessiana evaluada en el punto crítico en cuestión alternen en signo empezando por signo negativo. Para el caso de un mínimo se requiere que todos sean positivos.

Caso n variables:

En esta situación el problema se presenta como:

$$\boxed{\underset{x_1, x_2, \ldots, x_n}{Max} \quad f = f(x_1, x_2, \ldots, x_n)}$$

donde ahora la condición de primer orden viene dada por es siguiente sistema de n ecuaciones:

$$\begin{cases} f_{x_1}(x_1, x_2, \ldots, x_n) = 0 \\ f_{x_2}(x_1, x_2, \ldots, x_n) = 0 \\ \vdots \\ f_{x_n}(x_1, x_2, \ldots, x_n) = 0 \end{cases} \quad \Rightarrow (x_1^*, x_2^*, \ldots, x_n^*)$$

Donde al igual que en el caso anterior el sistema es no lineal y puede presentar múltiples soluciones. Cada una de ellas será un punto crítico y para decidir si dichos puntos son o no puntos óptimos se recurre a analizar el signo de los menores principales de la matriz hessiana evaluada en cada punto critico:

$$\mathbf{H}_f(x_1^*, x_2^*, ..., x_n^*) = \begin{pmatrix} f_{x_1x_1}(x_1^*, x_2^*, ..., x_n^*) & \dots & f_{x_1x_n}(x_1^*, x_2^*, ..., x_n^*) \\ \vdots & \ddots & \vdots \\ f_{x_nx_1}(x_1^*, x_2^*, ..., x_n^*) & \cdots & f_{x_nx_n}(x_1^*, x_2^*, ..., x_n^*) \end{pmatrix}$$

En esta situación para el caso de mínimo es suficiente que todos los menores principales sean positivos mientras que para el caso de máximo se requiere que alternen en signo comenzando por signo negativo. Algebraicamente se tiene:

$$Mínimo \begin{cases} |\mathbf{H}_1| > 0 \\ |\mathbf{H}_2| > 0 \\ \vdots \\ |\mathbf{H}_n| > 0 \end{cases} \qquad Máximo \begin{cases} |\mathbf{H}_1| < 0 \\ |\mathbf{H}_2| > 0 \\ \vdots \\ (-1)^n |\mathbf{H}_n| > 0 \end{cases}$$

4- <u>OPTIMIZACIÓN RESTRICCIONES DE IGUALDAD</u>

Caso 2 variables y una restricción:

Este tipo de problema consiste en hallar los valores de x_1 y x_2 que perteneciendo a una curva del dominio ($g(x_1, x_2) = m$) confieran a f un valor máximo o mínimo. Formalmente este problema puede escribirse de la siguiente manera:

$$\boxed{\begin{array}{l} \underset{x_1, x_2}{Max}\ f = f(x_1, x_2) \\ st: \quad g(x_1, x_2) = m \end{array}}$$

Para su resolución se hace uso de la función Lagrangeana (L) y de los multiplicadores de Lagrange (ë) como sigue:

$$\underset{x_1, x_2}{Max} L = f(x_1, x_2) + l[m - g(x_1, x_2)]$$

La condición de primer orden ahora se escribe en términos de la función Lagrangeana y arroja el siguiente sistema de ecuaciones:

$$\begin{cases} L_{x_1}(x_1, x_2) = 0 \\ L_{x_2}(x_1, x_2) = 0 \Rightarrow (x_1^*, x_2^*, l^*) \\ L_l(x_1, x_2) = 0 \end{cases}$$

Al igual que en los otros casos las ecuaciones pueden resultar ser no lineales y con soluciones múltiples. Cada una de dichas soluciones constituye los puntos críticos que luego deberán verificar las condiciones de segundo orden. En este caso se recurre a una matriz hessiana de la función Lagrangeana orlada con ceros y evaluada en cada punto crítico. Luego sobre estas matrices se toman los menores principales:

$$\bar{\mathbf{H}}_L(x_1^*, x_2^*, l^*) = \begin{pmatrix} 0 & g_{x_1}(x_1^*, x_2^*, l^*) & g_{x_2}(x_1^*, x_2^*, l^*) \\ g_{x_1}(x_1^*, x_2^*, l^*) & L_{x_1x_1}(x_1^*, x_2^*, l^*) & L_{x_1x_2}(x_1^*, x_2^*, l^*) \\ g_{x_2}(x_1^*, x_2^*, l^*) & L_{x_2x_1}(x_1^*, x_2^*, l^*) & L_{x_2x_2}(x_1^*, x_2^*, l^*) \end{pmatrix}$$

$$Mínimo: \left|\mathbf{H}_3\right| < 0 \qquad Máximo: \left|\mathbf{H}_3\right| > 0$$

Como se aprecia los menores principales se toman a partir del orden tres por lo que en definitiva solo hay que atender al signo del determinante de la matriz hessiana orlada evaluada en cada punto crítico.

Caso n variables y m restricciones (m < n)

Similarmente al caso planteado con anterioridad, el problema consiste en hallar un punto n dimensional (es decir los valores de $x_1, x_2, \ldots, x_n$) tales que satisfaciendo un conjunto de restricciones (siempre menor al número de incógnitas) confieran a f un valor máximo o mínimo. Formalmente el problema se traduce a:

$$\boxed{\begin{aligned} &\underset{x_1, x_2, \ldots, x_n}{Max} \quad f = f(x_1, x_2, \ldots, x_n) \\ &st: \quad g_1(x_1, x_2, \ldots, x_n) = m_1 \\ &\qquad\;\; g_2(x_1, x_2, \ldots, x_n) = m_2 \\ &\qquad\;\; \vdots \\ &\qquad\;\; g_m(x_1, x_2, \ldots, x_n) = m_m \end{aligned}}$$

Nuevamente se plantea la función Lagrangeana que ahora se generaliza de la siguiente manera:

$$\underset{x_1, x_2, \ldots, x_n}{Max} \; L = f(x_1, x_2, \ldots, x_n) + l_1[m_1 - g_1(x_1, x_2, \ldots, x_n)] + l_2[m_2 - g_2(x_1, x_2, \ldots, x_n)] + \cdots + l_m[m_m - g_m(x_1, x_2, \ldots, x_n)]$$

Mientras que la condición de primer orden arroja el siguiente sistema de $n + m$ ecuaciones

$$\left\{\begin{array}{l} L_{x_1}(x_1,x_2)=0 \\ L_{x_2}(x_1,x_2)=0 \\ \qquad\vdots \\ L_{x_n}(x_1,x_2)=0 \\ L_{l_1}(x_1,x_2)=0 \\ L_{l_2}(x_1,x_2)=0 \\ \qquad\vdots \\ L_{l_m}(x_1,x_2)=0 \end{array}\right. \Rightarrow (x_1^*, x_2^*, ..., x_n^*,\ l_1^*, l_2^*, ...,\ l_m^*)$$

Las soluciones para dicho sistema constituyen los puntos críticos candidatos a extremos relativos (máximos o mínimos) de la función sujeta a las restricciones. A continuación se debe recurrir a la matriz hessiana orlada que para el caso de m restricciones adopta la siguiente forma en términos matriciales:

$$\bar{\mathbf{H}}_L(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*) = \begin{pmatrix} \mathbf{0}_{m\times m} & \mathbf{g_X}(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*)_{m\,x\,n} \\ \mathbf{g_X}^T(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*)_{m\,x\,n} & \mathbf{H}_L(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*)_{n\,x\,n} \end{pmatrix}$$

donde:

$\mathbf{0}_{m\,x\,m}$ es una matriz cuadrada de ceros de orden *m x m*

$$\mathbf{g_X}(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*)_{m\,x\,n} = \begin{pmatrix} g^1{}_{x_1}(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*) & \cdots & g^1{}_{x_n}(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*) \\ \vdots & \ddots & \vdots \\ g^m{}_{x_1}(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*) & \cdots & g^m{}_{x_n}(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...l_m^*) \end{pmatrix}$$

$\mathbf{g_X}^T(x_1^*, x_2^*, ..., x_n^*,\ l_1^*, ...\ l_m^*)_{m\,x\,n}$ es la transpuesta de la matriz anterior

Una vez obtenida la matriz hessiana orlada se debe atender al signo de los menores principales a partir del orden *2 m +1*. Para el caso de mínimo se requieren que todos tengan el signo de $(-1)^m$ y para máximo se requiere que los menores alternen en signo comenzando por el signo de $(-1)^{m+1}$.

5- OPTIMIZACIÓN RESTRICCIONES DE DESIGUALDAD

En este caso el problema es:

$$\boxed{\begin{array}{l} \underset{x_1,x_2,...,x_n}{Max}\ f = f(x_1, x_2, ..., x_n) \\ st:\ \ g_1(x_1, x_2, ..., x_n) < m_1 \\ \qquad g_2(x_1, x_2, ..., x_n) < m_2 \\ \qquad \vdots \\ \qquad g_m(x_1, x_2, ..., x_n) < m_m \end{array}}$$

La función Lagrangeana será:

$$\underset{x_1,x_2,\ldots,x_n}{Max} \; L = f(x_1,x_2,\ldots,x_n) + l_1[m_1 - g_1(x_1,x_2,\ldots,x_n)] + l_2[m_2 - g_2(x_1,x_2,\ldots,x_n)] + \cdots + l_m[m_m - g_m(x_1,x_2,\ldots,x_n)]$$

Las condiciones de Kuhn-Tucker para máximo son:

$$x_j \frac{\partial L}{\partial x_j} = 0 \qquad \frac{\partial L}{\partial x_j} \le 0 \qquad x_j \ge 0 \qquad (j = 1,2,\ldots,n)$$

$$l_i \frac{\partial L}{\partial l_i} = 0 \qquad \frac{\partial L}{\partial l_i} \ge 0 \qquad l_i \ge 0 \qquad (i = 1,2,\ldots,m)$$

Los problemas de minimización pueden resolverse en términos de las condiciones para máximo mediante el simple artilugio:

- Primero se debe definir la función objetivo como la opuesta (es decir multiplicada por -1)
- Los problemas de minimización suelen tener desigualdades del tipo mayor o igual en consecuencia se las debe transformar el sentido de las desigualdades multiplicando ambos miembros por -1 de modo que las desigualdades sean del tipo menor o igual

6- IMPLEMENTACIÓN EN MAPLE

Como se apreció en las secciones anteriores la tarea de resolver los sistemas de ecuaciones que surgen de las condiciones de primer orden como así también el cómputo de las matrices hessianas, su evolución en los puntos críticos y el cálculo de los menores principales, es realmente tediosa y demandante de un alto insumo de tiempo. En este sentido el presente paquete de optimización permite al usuario mediante sencillos comandos resolver las condiciones de primer y segundo orden para todos los puntos críticos de los diversos problemas planteados en la sección anterior. Posee además la generalidad de ser extensivo a n variables y m restricciones.
Para ello se han creado un conjunto de rutinas que permiten incorporar nuevas funciones al lenguaje Maple

Los comandos disponibles son los siguientes:
libre: este comando permite obtener cada uno de los puntos críticos de un problema de optimización libre en n variables, el cómputo de la matriz hessiana evaluada en cada uno de los puntos críticos y el cálculo de los menores principales asociados a cada candidato a óptimo.
lagrange: por medio de esta orden se obtienen los candidatos a óptimos (incluyendo a los multiplicadores de Lagrange) de un problema de optimización restringida en n variables y m restricciones como así también las matrices hessianas orladas evaluadas en sus respectivos puntos críticos y los correspondientes menores principales a partir del de orden 2m+1
Maxkt: este comando habilita al usuario a hallar los puntos de un problema de programación no lineal en n variables y m restricciones que satisfacen las condiciones de Kuhn-Tucker para el caso de maximización. Mediante un simple artilugio este comando también puede ser usado para el caso de minimización.

INICIALIZACIONES :

A continuación se exponen el conjunto de rutinas que permiten al programa reconocer los nuevos comandos anteriormente enunciados.

```
# Rutina para resolver problemas de optimización libre #

# Creación de la function libre #
restart;
libre:=proc(f,X)
local A, j, k, B;

# Condición de primer orden  #
print(____________________________________________________________);
_EnvExplicit := true:
A:=[solve({seq(diff(f,X[i])=0,i=1..nops(X))},{seq(X[i],i=1..nops(X))})]:
for j from 1 by 1 to nops(A) do
print(Punto_Critico[j]);

# Condición de Segundo orden #
print([A[j],fvalue=eval(f,A[j])]);
print(Hessian[j]=eval(linalg[hessian](f,X),A[j]));
B:= eval(linalg[hessian](f,X),A[j]):
print(Minors[j]=[seq(linalg[det](linalg[submatrix](B, 1..k, 1..k)),k=1..nops(X))]);
print(____________________________________________________________);

od;
end:
```

```
# Rutina para resolver problemas de optimización restricciones de igualdad #

# Creación de la function lagrange #
lagrange:=proc(h,g,v,l)
local A, j, k, B, f, T, X, m, q, G, GT, Ceros, HH;
f:= h-linalg[multiply](linalg[transpose](l),g):
T:= convert(linalg[stackmatrix](linalg[transpose]([v]),linalg[transpose]([l])),vector):
X:=[seq(T[m],m=1..(nops(v)+nops(l)))]:

# Condición de primer orden  #
print(____________________________________________________________);
_EnvExplicit := true:
A:=[solve({seq(diff(f,X[i])=0,i=1..nops(X))},{seq(X[i],i=1..nops(X))})]:
for j from 1 by 1 to nops(A) do
print(Punto_Critico[j]);
print(A[j]);
print(fvalue=eval(f,A[j]));

# Condición de Segundo orden #
G:=convert([seq([seq(diff(g[j],v[i]),i=1..nops(v))],j=1..nops(g))],matrix):
Ceros:=convert([seq([seq(0,i=1..nops(g))],j=1..nops(g))],matrix):
```

```
GT:=linalg[transpose](G):
HH:=linalg[hessian](f,v):
print(Hessian[j]=eval(linalg[stackmatrix](linalg[concat](Ceros,G),linalg[concat](GT,HH)),A[j
]));
B:= eval(linalg[stackmatrix](linalg[concat](Ceros,G),linalg[concat](GT,HH)),A[j]):
q:= 2*nops(l)+1:
print(Minors[j]=[seq(linalg[det](linalg[submatrix](B, 1..k, 1..k)),k=q..nops(X))]);
print(____________________________________________________________);
od;


end:
```

```
# Rutina para resolver problemas de optimización restricciones de desigualdad #

# Creación de la function Maxkt #
Maxkt:=proc(h,g,v,l)
local A, j, k, B, f, T, X, m, q, qq, t, u, uu, gg, uuu, L;
f:= h-linalg[multiply](linalg[transpose](l),g):
T:= convert(linalg[stackmatrix](linalg[transpose]([v]),[l]),vector):
X:=[seq(T[m],m=1..(nops (v)+nops(l)))]:
L:=[]: print(____________________________________________________________);

# Condición de primer Kuhn-Tucker  #
_EnvExplicit := true:
A:=[solve({seq(X[i]*diff(f,X[i])=0,i=1..nops(X))},{seq(X[i],i=1..nops(X))})]:
gg:=[]:
for j from 1 by 1 to nops(A) do
qq:=0:
for t from 1 by 1 to nops(v) do
if(evalf(eval(diff(f,v[t]),A[j]))<=0) then qq:=qq+1 fi:
od;
for u from 1 by 1 to nops(l) do
if(evalf(eval(diff(f,l[u]),A[j]))>=0) then qq:=qq+1 fi:
od;
for uu from 1 by 1 to nops(X) do
if(evalf(eval(X,A[j]))[uu]>=0) then qq:=qq+1 fi:
od;

if(qq=2*nops(X)) then
print(Punto_Critico[j]);
print(A[j]);
print(fvalue=eval(f,A[j]));
gg:= [op(gg),eval(f,A[j])]:
L := [op(L),A[j]]:
print(____________________________________________________________);
fi;

od;
```

```
for uuu from 1 by 1 to nops(gg) do
if(max(seq(gg[ii],ii=1..nops(gg)))=gg[uuu]) then print([MaxGlobal_fvalue=gg[uuu],L[uuu]])
fi;
od;

end:
```

7- SINTAXIS Y EJEMPLOS

OPTIMIZACIÓN LIBRE

Sintaxis

libre (*función* **,** *listado de variables independientes*);

Ejemplo

Sea el siguiente problema: ***Max f = x^3 + 3*x*y^2 - x*y***

```
> libre(x^3+3*x*y^2-x*y,[x,y]);
```

$$Punto\_Critico_1$$

$$[\{y=0, x=0\}, fvalue = 0]$$

$$Hessian_1 = \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix}$$

$$Minors_1 = [0, -1]$$

$$Punto\_Critico_2$$

$$\left[\{x=0, y=\frac{1}{3}\}, fvalue = 0\right]$$

$$Hessian_2 = \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

$$Minors_2 = [0, -1]$$

$$Punto\_Critico_3$$

$$\left[\{y = \frac{1}{6}, x = \frac{1}{6}\}, fvalue = \frac{-1}{108} \right]$$

$$Hessian_3 = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

$$Minors_3 = [1, 1]$$

$$Punto\_Critico_4$$

$$\left[\{y = \frac{1}{6}, x = \frac{-1}{6}\}, fvalue = \frac{1}{108} \right]$$

$$Hessian_4 = \begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$$

$$Minors_4 = [-1, 1]$$

OPTIMIZACIÓN RESTRICCIONES DE IGUALDAD

Sintaxis

lagrange (*función* **,** *listado de restricciones igualadas a cero, listado de variables independientes, listado de multiplicadores de lagrange*);

Ejemplo

Sea el siguiente problema: ***Max f = x^2+y^2***

st: 3x+5y=1

```
> lagrange(x^2+y^2,[3*x+5*y-1],[x,y],[l]);
```

$$Punto\_Critico_1$$

$$\{x = \frac{3}{34}, y = \frac{5}{34}, l = \frac{1}{17}\}$$

$$fvalue = \frac{1}{34}$$

$$Hessian_1 = \begin{bmatrix} 0 & 3 & 5 \\ 3 & 2 & 0 \\ 5 & 0 & 2 \end{bmatrix}$$

$$Minors_1 = [-68]$$

OPTIMIZACIÓN RESTRICCIONES DE DESIGUALDAD

Sintaxis

Maxkt (*función* **,** *listado de restricciones igualadas a cero "del tipo menor que"* , *listado de variables independientes, listado de multiplicadores de lagrange*);

Ejemplo

Sea el siguiente problema: ***Max f* = *x*^ 2 + *y*^ 2**

***st:* 3 *x* + 5 *y* < 1**

```
> Maxkt(x^2+y^2,[3*x+5*y-1],[x,y],[l]);
```

$$Punto\_Critico_1$$

$$\{l = 0, y = 0, x = 0\}$$

$$fvalue = 0$$

$$Punto\_Critico_2$$

$$\{l = 0, y = 0, x = 0\}$$

$$fvalue = 0$$

$$Punto\_Critico_3$$

$$\{x = 0, l = \frac{2}{25}, y = \frac{1}{5}\}$$

$$fvalue = \frac{1}{25}$$

$$Punto\_Critico_4$$

$$\{y = 0, x = \frac{1}{3}, l = \frac{2}{9}\}$$

$$fvalue = \frac{1}{9}$$

$$Punto\_Critico_5$$

$$\{x = \frac{3}{34}, y = \frac{5}{34}, l = \frac{1}{17}\}$$

$$fvalue = \frac{1}{34}$$

$$\left[MaxGlobal\_fvalue = \frac{1}{9}, \{y = 0, x = \frac{1}{3}, l = \frac{2}{9}\} \right]$$

8- BIBLIOGRAFÍA

- **APÓSTOL, Tom:** *Calculus.* Ed. Reverté. 1976. Segunda Edición
- **CHIANG, Alpha:** *Economía Matemática.* Ed. McGraw-Hill. 1998
- **LUEMBERGER, M.:** *Programcicion lineal y no lineal*
- **STEIN, S.:** *Cálculo con geometría analítica. Ed Mc. Graw-Hill. 1995*
- **STEWARD, James:** *Cálculo.* Ed. Iberoamerica.1981